/*

Laura Richman
Proposal title: Draining Social Ties and Financial Decision Making

HYPOTHESES

Stated-Hyp 1: Main effect of draining ties on a higher willingness to engage in risky financial decisions.

	Test-Hyp 1: Participants in the draining ties condition will be less future oriented than participants in the supportive ties condition.
	Test-Hyp 2: Participants in the draining ties condition will make more risky decisions than participants in the supportive ties condition.

Stated-Hyp 2: The effect of draining ties on a higher willingness to engage in risky financial decisions is moderated by SES such that the effects will be stronger for low than high SES participants. 

	Test-Hyp 3: The effect of the ties condition on future orientations will be moderated by SES.
	Test-Hyp 4: The effect of the ties condition on risky financial decisions will be moderated by SES.
		
Stated-Hyp 3: We predict that supportive ties will not have an interaction effect with SES, but will be equally beneficial (not lead to lower future orientation or risky decisions) for both high and low SES groups.

	This stated hypothesis could not be converted into test hypothesis(es).
		
********************************************************************************

NOTES: 
	
1. Unclear if some items for the risky decisions scale should be reverse-coded.
However, because none of the items correlated negatively with the rest of the 
scale, we did not reverse-code any of the items.
2. Unclear if the main effect hypothesis applies to both DVs
	
*/

clear all
use Richman839, clear

********************************************************************************

* RECODING

* experiment condition
	tab xtess227, mis
	tab xtess227, nolabel
	
	gen sesCondition = 0
	replace sesCondition = 1 if xtess227 == 3
	replace sesCondition = 1 if xtess227 == 4
	label define sesCondition_label 0 "Low" 1 "High"
	label values sesCondition sesCondition_label  	
	tab xtess227 sesCondition
	
	gen tieCondition = 0
	replace tieCondition = 1 if xtess227 == 1
	replace tieCondition = 1 if xtess227 == 3
	label define tieCondition_label 0 "Supportive" 1 "Draining"
	label values tieCondition tieCondition_label
	tab xtess227 tieCondition
	
* DVs
	tab b01_1, mis
	tab b01_1, nolabel
	tab b01_2, nolabel
	tab b01_3, nolabel
	tab b01_4, nolabel
	tab b01_5, nolabel
	replace b01_1 = . if b01_1 == -1
	replace b01_2 = . if b01_2 == -1
	replace b01_3 = . if b01_3 == -1
	replace b01_4 = . if b01_4 == -1
	replace b01_5 = . if b01_5 == -1
	recode b01_1 (4=1) (3=2) (2=3) (1=4)
	recode b01_3 (4=1) (3=2) (2=3) (1=4)
	recode b01_5 (4=1) (3=2) (2=3) (1=4)
	gen futori = b01_1 + b01_2 + b01_3 + b01_4 + b01_5
	tab futori, mis

	tab b03_1, mis
	tab b03_1, nolabel
	tab b03_2, nolabel
	tab b03_3, nolabel
	tab b03_4, nolabel
	tab b03_5, nolabel
	tab b03_6, nolabel
	replace b03_1 = . if b03_1 == -1
	replace b03_2 = . if b03_2 == -1
	replace b03_3 = . if b03_3 == -1
	replace b03_4 = . if b03_4 == -1
	replace b03_5 = . if b03_5 == -1
	replace b03_6 = . if b03_6 == -1
	alpha b03_1 b03_2 b03_3 b03_4 b03_5 b03_6, item
	gen risdec = b03_1 + b03_2 + b03_3 + b03_4 + b03_5 + b03_6
	tab risdec, mis

********************************************************************************

* ANALYSIS

* Test-Hyp 1: Participants in the draining ties condition will be less future oriented than participants in the supportive ties condition.

	reg futori i.tieCondition
	// no support for H1 (p = .455)
	tess 1.tieCondition -, init(Richman839) bonf(2)
	
* Test-Hyp 2: Participants in the draining ties condition will make more risky decisions than participants in the supportive ties condition.

	reg risdec i.tieCondition
	// no support for H2 (p = .611)
	tess 1.tieCondition +, bonf(2)
	
* Test-Hyp 3: The effect of the ties condition on future orientations will be moderated by SES.

	reg futori i.tieCondition##i.sesCondition
	// no support for H3 (p = .107)
	tess 1.tieCondition#1.sesCondition, bonf(2)
	
* Test-Hyp 4: The effect of the ties condition on risky financial decisions will be moderated by SES.

	reg risdec i.tieCondition##i.sesCondition
	// no support for H4 (p = .648)
	tess 1.tieCondition#1.sesCondition, bonf(2)
